Feature Measurement Tool

Jessica Leonard
GEOG 590

Purpose

- Create toolbar to assist in tax parcel mapping
  - Update Acreage
    - When selected, acreage field will be added to the attribute table and acreage for all polygons will be updated
  - Calculate Features
    - User is able to select single or multiple polygons and a message box opens displaying the acreage

Classes & Interfaces

IFeature	‘access to return & set properties of features
IFeatureClass	‘control features & behavior of feature class
IFeatureLayer	‘control aspects of feature layer
IFeatureCursor	‘provides access to fields
IFeatureSelection	‘access members that control selection
IGeometry	‘properties & behavior of geometric objects
IArea	‘return polygon properties
IEnumLayer	‘allows iteration through layers

Algorithms

- Acreage = Area/43,560

Update Acreage

```
Private Sub UpdateAcreage_Select()
'This VBA script is to calculate the Area of polygons and add the area as new field
'in the attribute table of the polygon
   Dim pMxDoc As IMxDocument
   Set pMxDoc = ThisDocument

   'Make sure a map is active for program to continue
   If Not TypeOf pMxDoc.ActiveView Is IMap Then
      MsgBox "A map must be active!"
      Exit Sub
   End If

   Dim pFLayer As IFeatureLayer
   Dim pFClass As IFeatureClass

   'Ensure that a polygon shapefile has been selected for acreage to be calculated
   Set pFLayer = pMxDoc.SelectedLayer
   If pFLayer Is Nothing Then
      MsgBox "A single polygon shapefile must be selected!"
      Exit Sub
   End If
   Set pFClass = pFLayer.FeatureClass

   If pFClass.ShapeType <> esriGeometryPolygon Then
      MsgBox "You must Select a single polygon shapefile!"
      Exit Sub
   End If
```

Update Acreage (2)

```
Dim indexA As Long
   indexA = pFClass.FindField("Acreage")
   If indexA < 0 Then 'If there is no the field of ACREAGE
      Dim pFieldx As IFieldEdit
      Set pFieldx = New Field

      With pFieldx
         .Type = esriFieldTypeDouble
         .Name = "Acreage"
      End With
      pFClass.AddField pFieldx
   End If

   indexA = pFClass.FindField("Acreage")

   'Get a cursor that can be used to update features for all records in polygon feature
   Dim pFCursor As IFeatureCursor
   Set pFCursor = pFClass.Update(Nothing, False) 'returns all records

   Dim pFeature As IFeature
   Set pFeature = pFCursor.NextFeature ' move to first feature
```

Update Acreage (3)

```
    Dim pShape As IGeometry
    Dim pArea As IArea

    Dim aarea As Double

    While Not pFeature Is Nothing
        Set pShape = pFeature.Shape
        Set pArea = pShape
        aarea = pArea.Area / 43560
        pFeature.Value(indexA) = aarea
        pFCursor.UpdateFeature pFeature
        Set pFeature = pFCursor.NextFeature
    Wend

    MsgBox "Acreage has been updated in your attribute table!"
End Sub
```

Calculate Features

```
Private Sub Calculate_Select()
    Dim pMxDoc As IMxDocument
    Set pMxDoc = ThisDocument

    Dim pUID As New UID
    'IGeoFeatureLayer IID
    pUID = "{E156D7E5-22AF-11D3-9F99-00C04F6BC78E}"

    Dim pEnumLayer As IEnumLayer
    Set pEnumLayer = pMxDoc.FocusMap.Layers(pUID, True)
    pEnumLayer.Reset

    Dim pFeatureLayer As IFeatureLayer
    Dim pFeatureSelection As IFeatureSelection
    Dim pFeatureCursor As IFeatureCursor
    Dim pFeature As IFeature
    Dim pArea As IArea
    Dim dblTotalArea As Double
```

Calculate Features (2)

```
'Loop through selected polygons and extract area
Set pFeatureLayer = pEnumLayer.Next
Do Until (pFeatureLayer Is Nothing)
If (pFeatureLayer.FeatureClass.ShapeType = esriGeometryPolygon) Then
   Set pFeatureSelection = pFeatureLayer

   If (pFeatureSelection.SelectionSet.Count <> 0) Then
      pFeatureSelection.SelectionSet.Search Nothing, True, pFeatureCursor
      Set pFeature = pFeatureCursor.NextFeature

      Do Until (pFeature Is Nothing)
         Set pArea = pFeature.Shape
         dblTotalArea = dblTotalArea + pArea.Area
         Set pFeature = pFeatureCursor.NextFeature
      Loop
   End If
End If
Set pFeatureLayer = pEnumLayer.Next
Loop
```

Calculate Features (3)

```
'Convert area to acreage
 Dim dblAcreage As Double
 dblAcreage = dblTotalArea / 43560

 MsgBox "Total area for selected polygon features = "  &
  Round(dblTotalArea, 2) & " feet" _
       & (Chr(13)) & "Total acreage for selected polygons = " &
  Round(dblAcreage, 2) & " acres"

End Sub
```

User's Guide

- Calculations for acreage assumes area is in feet
- Feature layer must first be selected in order for update acreage tool to run
- Must first select polygon features for the calculate tool, either by query or using the select features tool

Sources

- ArcScripts
  - Borrowed and modified some parts of code from Neil Banerjee and Duane Wilkens